OCS STUDY
MMS 85-0046

A NUMERICAL MUD DISCHARGE PLUME MODEL FOR OFFSHORE DRILLING OPERATIONS

1985

U.S. DEPARTMENT OF THE INTERIOR/MINERALS MANAGEMENT SERVICE

This report has not been edited for conformity with MMS editorial standards.

OCS STUDY
MMS 85-0046

A NUMERICAL MUD DISCHARGE PLUME MODEL FOR OFFSHORE DRILLING OPERATIONS

by

Roger Multer

Hydraulics Laboratory
U.S. Army Engineer Waterways Experiment Station
P.O. Box 631
Vicksburg, Mississippi 39180

Prepared under Contract 14-12-0001-30012

for

Minerals Management Service
Gulf of Mexico OCS Regional Office
Metairie, Louisiana 70010

July 1985

REPORT AVAILABILITY

Preparation of this report was conducted under Contract between the Minerals Management Service and the U.S. Army Corps of Engineers. Extra copies of the report may be obtained from the Public Information Section (Mail Stop OPS-3-4) at the following address:

Minerals Management Service
Gulf of Mexico OCS Regional Office
U.S. Department of the Interior
P.O. Box 7944
Metairie, Louisiana 70010
Attn: Public Information Section
Phone: (504) 838-0519

DISCLAIMER

This report has been reviewed by the Minerals Management Service and approved for publication. Approval does not signify that the contents necessarily reflect the views and policies of the Service, nor does mention of trade names or commercial products constitute endorsement or recommendation for use.

This volume should be cited as:

Multer, R. 1985. A Numerical Mud Discharge Plume Model for Offshore Drilling Operations. (Final Report by U.S. Army Corps of Engineers submitted to Minerals Management Service, Metairie, LA. Contract No. 14-12-0001-30012). 16p.

ABSTRACT

This study involved the modification and subsequent application of a generic plume model developed by the Walden Division of Abcor Inc. under contract to the US Army Corps of Engineers. The mathematical basis of the model is a unidirectional, steady, Reynolds type diffusion equation. To close the governing equation, coefficients of eddy diffusion are introduced and a hypothesis of similar mass and momentum diffusion invoked to define them. Also made is an assumption that the velocity may be replaced by its mean over the depth. This substitution facilitates separation of the original equation into two partial differential equations. One equation models the effect of lateral diffusion and is solved analytically. The other models the interaction of settling, longitudinal transport, and vertical diffusion and is solved numerically. The effect of this separation of variables is a substantial reduction in the computational labor.

As a part of the study a new computer program was written to perform the numerical computations. The reasons for this were

1) The original model only allowed for initial, upstream distributions which were uniform over the depth. The new model allows for inflow of the material through a "window" which begins at some elevation at or above the bottom and ends at or below the free-surface.

2) The original model required that the distribution of material at the upstream boundary be specified as concentration. The revised model allows either concentration or mass flux to be used.

3) The original model used (apparently) a general library subroutine for solving systems of numerical equations. It was desired to have a version of the program which could be used reasonably on much slower microcomputers. Therefore, a semi-implicit, tridiagonal algorithm was introduced to make the computations highly efficient.

4) Output tables not generated by the original model were desired and code appropriate for generating these was added. Principal among these is a table which gives the rate of deposition.

A second program which sets up the data needed by the numerical computation program was also written. This program asks the modeler a series of questions and stores the responses in a data file which is subsequently used by the numerical plume model. The combination of the two programs form an interactive, easy to use, tool for studying mud plumes.

PREFACE

The study reported herein was conducted during the period March 1983 to August 1984 by the Hydraulics Laboratory of the US Army Engineers Waterways Experiment Station (WES) under the general supervision of Messrs. H. B. Simmons and F. A. Herrmann, Jr., Chiefs of the Hydraulics Laboratory, and M. B. Boyd, Chief of the Hydraulic Analysis Division. The study was sponsored by the Mineral Management Service (MMS) of the Department of Interior under Requisition Numbers 3-6036-0010 and 3-6036-0046, and Contract Number 14-12-001-30012.

Dr. B. H. Johnson was project manager for the study. Dr. R. H. Multer developed the computer code and conducted the numerical study. Commander and Director of WES during the period of this study was COL Tilford C. Creel, CE Technical Director was Mr. Fred R. Brown.

CONTENTS

PART I: INTRODUCTION

1. The study which is described in this report involved the modification of a generic plume model to compute the dispersion of material discharged in off-shore drilling operations. The original model was developed by B. A. Wechsler and D. A. Cogley* under contract to the Corps of Engineers as a part of the Dredged Material Research Program. The model assumes that the velocity field is time-invariant and constant in magnitude and direction. The depth of the fluid is also constant. The planer region occupied by the fluid is assumed to be a half-plane. This latter assumption is consistent with the environment of off-shore drilling rigs which are usually far from shoreline boundaries.

The objectives of the study were:

- a. Make the model more flexible so that cases where the initial distribution of sediment varied with respect to depth could be considered.
- b. Incorporate a nonconstant definition of eddy diffusion into the model.
- c. Incorporate an algorithm into the model to compute the total sediment flux at various downstream distances.
- d. Incorporate an algorithm into the model to compute the rate of deposition.
- e. Adapt the model for execution on small micro- or mini-computers.

These are in essence the tasks which were accomplished. Relative to item (e), an examination of the model was made in terms of its numerical efficiency. It was determined that the efficiency of the computations could be much improved by taking advantage of the fact that the system of equations to be solved was tridiagonal and stationary so that the equation needed to be solved only once instead of for each progressive step downstream from the injection point.

---

* Wechsler, B. A. and Cogley, D. R., "A Laboratory Study of the Turbidity Generation Potential of Sediments to be Dredged," WES Technical Report D-77-14, November 1977.

2. Emperical relations are required to close the analytical description of a turbulent-plume. Wechsler and Cogley use an algebraic closure which is not unreasonable, although, conversely it is not necessarily correct. Pursuit of this point is beyond the scope of the current investigation.

PART II: DESCRIPTION OF THE MODEL

Basic Equations

3. The basic equation to be solved is the steady state three-dimensional transport-diffusion equation. Assuming eddy diffusion, this equation can be written as:

$$\underbrace{\frac{\partial}{\partial x}(uc)}_{\text{downstream advection}} + \underbrace{\frac{\partial}{\partial v} \int wf(w)dw}_{\text{vertical sedimentation}} - \underbrace{\frac{\partial}{\partial x}\left(E_x \frac{\partial c}{\partial x}\right) - \frac{\partial}{\partial y}\left(E_y \frac{\partial c}{\partial y}\right) - \frac{\partial}{\partial z}\left(E_z \frac{\partial c}{\partial z}\right)}_{\text{eddy diffusion}} = 0 \qquad (1)$$

where

x = downstream coordinate

y = vertical coordinate

z = lateral coordinate

u = current velocity at any point

c = sediment concentration

w = settling velocity

$f(w)$ = settling-velocity frequency distribution

E_x, E_y, E_z = eddy diffusivities in x, y, and z directions

4. This equation is based on the assumptions that the flow is steady, uniform, and fully turbulent, and that eddy diffusion can be characterized by Fick's law with eddy diffusion coefficients. Equation 1 is unclosed and several assumptions were made by Cogley and Wechsler in order to simplify its solution. These are:

a. Eddy diffusion in the downstream direction is negligible compared to the other two diffusive transport terms.

b. The flow is fully turbulent, therefore, the local velocity u in Equation 1 can be replaced by the mean velocity U.

c. The mass and momentum diffusion coefficients in the vertical direction are essentially the same.

d. The shear stress is linearly distributed over the depth, and the velocity distribution is given by Von Karman's "universal" distribution.

e. The lateral eddy diffusion coefficient is essentially constant over the depth.

5. Based on the above assumptions, algebraic expressions for E_y and E_z can be written as:

$$E_y = 0.4\ U\ y(1 - y/h)\ \mathrm{sqrt}(f/8) \tag{2}$$

and

$$E_z = 0.2\ h\ U\ \mathrm{sqrt}(f/8) \tag{3}$$

where f is the Darcy friction factor and U is the depth averaged current velocity. For each fraction of the sediment load for which the settling velocity is constant, Equation 1 may be rewritten as:

$$U \frac{\partial C}{\partial x} + w \frac{\partial C}{\partial y} - \frac{\partial}{\partial y} E_y \frac{\partial C}{\partial y} - \frac{\partial}{\partial z} E_z \frac{\partial C}{\partial z} = 0 , \tag{4}$$

which together with Equations 2 and 3 is the model of the process considered here.

Boundary Conditions

6. Physically, the boundary condition for the free surface must specify that there is no flux across the free surface. This condition is described mathematically by:

$$E_y \frac{\partial C}{\partial y} + wC = 0 \tag{5}$$

Cogley and Wechsler assume that all of the sediment reaching the bottom is deposited and that there is no reentrainment. The corresponding boundary condition is:

$$E_y \frac{\partial C}{\partial y} = 0 \tag{6}$$

Separation of Variables

7. It may be assumed that the solution for Equation 4 is of the form:

$$C(x,y,z) = C_o(x,y)\ C_d(x,z) \tag{7}$$

On substituting Equation 7 into Equation 4 the following two equations are obtained:

$$U \frac{\partial C_o}{\partial x} + w \frac{\partial C_o}{\partial y} = \frac{\partial}{\partial y}\left(E_y \frac{\partial C_o}{\partial y}\right) \tag{8}$$

and

$$U \frac{\partial C_d}{\partial x} = \frac{\partial}{\partial z}\left(E_z \frac{\partial C_d}{\partial z}\right) \tag{9}$$

Or, stated another way, if C_o and C_d are solutions of Equations 8 and 9, respectively, their product is a solution of Equation 4. A solution of Equation 9, which is applicable to situations where there are no lateral constraints on the flow is:

$$C_d(x,z) = \frac{4\pi x E_z}{U}^{-1/2} \int_{-b}^{b} \exp\left(-\frac{(z-v)^2 u}{4xE_z}\right) dv \tag{10}$$

Equation 8 is a mathematical description of a two-dimensional diffusion process in which there is no lateral diffusion. It must be solved numerically.

8. The plume's suspended sediment concentration i.e. the solution given by Equation 7, is symmetric about the x-axis and the maximum concentration occurs at the centerline. Equation 9 models the distribution of the sediment in the lateral direction. The quantity C_d (x,0) is then in effect a description of the centerline dilution which results from lateral diffusion.

9. In concluding this section it seems pertinent to observe that Equation 4 is homogenous which means that if C(x,y,z) is some solution then so is aC(x,y,z) where a is any constant. The consequence of this homogenuity is that C has no particular dimensions associated with it. The convention which has been adopted is to specify distances in meters, velocities in meters/sec, and concentration in grams per cubic meter (equal to milligrams per liter). It follows that mass fluxes have the dimensions of grams/sec and that rates of deposition have the dimensions of grams/sec per square-meter.

Numerical Approximations

10. A numerical approximation of Equation 8 may be developed on a Cartesian grid. The approximation employed in this study is explained briefly in this section. Equation 8 may be rewritten as:

$$\frac{\partial}{\partial y}\left(E_y \frac{\partial C_o}{\partial y}\right) = U \frac{\partial C_o}{\partial z} + w \frac{\partial C_o}{\partial y} \tag{11}$$

Suppose first that there is no diffusion. The equation corresponding to this case is:

$$U \frac{\partial C_o}{\partial x} + w \frac{\partial C_o}{\partial y} = 0 \tag{12}$$

and it has the solution:

$$C_o(x,y) = f(y - \frac{w}{U} x) \tag{13}$$

The Cartesian grid (see Figure 1) may be defined by the generators $\{x(i+1) = x(i) + dx\}$ and $\{y(j+1) = y(j) + dy\}$ in which case (i,j) represents a unique grid point. According to Equation 13, C_o is constant along the line with slope w/U which passes thru the point (i,j). This line will intersect the vertical line $x(i-1)$ at an elevation of $y(j) - w/U\ dx$. Assuming that C_o is known at the set of points $\{i-1,\ j = 0,1,2,\dots\}$ and that w/U is less than dy/dx, C_o may be estimated by interpolating linearly between the points $(i-1,j)$ and $(i-1,j+1)$. The equation for this is:

$$C_o(i,j) = \frac{(dy + \frac{w}{U} dx)}{dy}\ C_o(i-1,j) - \frac{w}{U}\frac{dx}{dy}\ C_o(i-1,j+1) \tag{14}$$

Equation 14 is used as a replacement for the right hand side of Equation 11.

11. Using a Taylor series expansion it may be shown that to an order of dy squared:

$$\frac{\partial}{\partial y} E_y \frac{\partial C}{\partial y} = (E_y(j+1)+E_y(j))/(2*dy*dy)*C_o(i,j+1)$$

$$+ (E_y(j)+E_y(j-1))/(2*dy*dy)*C_o(i,j-1)$$

$$- (E_y(j+1)+2*E_y(j)+E_y(j-1))/(2*dy*dy)*C_o(i,j) \qquad (15)$$

This relationship is used as a replacement for the left hand side of Equation 11 in the numerical approximation.

12. The result of substituting Equations 14 and 15 into Equation 11 is a system of linear equations for each i. Examining Equation 15, it can be seen that the system of equations is tridiagonal and furthermore that the system is independent of i, i.e., x. These factors influence the time required to accomplish the computation since the system of equations need be solved only once. Also, while in general a computation of order n cubed is required to solve n linear equations, the computation required to solve a tridiagonal system is only of order $3n$.

PART III: APPLICATION OF THE MODEL

Model Input

13. The input quantities which must be specified are the depth h, the mean current velocity U, the Darcy friction factor f, and a description of the sediment and the manner in which it enters the flow. The sediment is regarded as consisting of a finite sum of parts or fractions, and the fall velocity for each fraction must be specified. The sediment enters the water column through an imaginary or virtual window (see Figure 1). The modeler, having specified the depth, h, specifies the number of stations in the y or vertical direction, Nys. Station Number 1 is at the bottom and the station Nys is at the free surface. The distance between stations in the vertical direction is

$$dh = h/(Nys - 1) \tag{16}$$

The bottom and top of the window are specified in terms of station numbers. For example, if

$$h = 50 \text{ meters}$$

and

$$Nys = 21$$

then

$$dh = 50/20 = 2.5 \text{ meters}$$

and if the bottom of the window is specified as vertical station 10 and the top of the window is specified as station 15 then

$$\text{elevation of window bottom} = 10*2.5 = 25 \text{ meters}$$

and

$$\text{elevation of top of window} = 15*2.5 = 37.5 \text{ meters}$$

Sediment inflow into the model may be specified in terms of either a mass concentration (mass/volume) or as a mass flux (mass/time).

14. Other quantities which the modeler must specify are the number of stations in the longitudinal direction, Nxs, and the distance, delx, between them; as well as, the number of stations in the lateral direction, Nzs, and the lateral distance between then, delz. These parameters may be selected as the modeler chooses, and in many cases the modeler may elect to make several runs with different selections of delx and delz.

15. Data for the model is setup by a program called FRNT. On typing the command FRNT into the microcomputer, the system responds with a series of questions:

```
STREAM VELOCITY (M/SEC) = ?
DEPTH IN METERS = ?
FRICTION FACTOR (RANGE = 0.005 TO 0.025) = ?
LONGITUDINAL DISTANCE BETWEEN STATIONS IN METERS = ?
LATERAL DIST. BETWEEN STATIONS IN METERS = ?
NUMBER OF STATIONS IN LONGITUDINAL DIRECTION = ?
NUMBER OF STATIONS IN THE VERTICAL DIRECTION = ?
NUMBER OF STATIONS IN THE LATERAL DIRECTION = ?
WIDTH OF INFLOW WINDOW (METERS) = ?
STATION NUMBER FOR BOTTOM OF WINDOW (1 IS BOTTOM) = ?
STATION NUMBER FOR TOP OF WINDOW (≤ NUMBER OF Ys) = ?
MASS FLUX OR CONC. BOUNDARY CONDITION ( M OR C) = ?
```

if response = M

then:

```
MASS FLUX (IN UNITS OF GRAMS/SEC) = ?
```

else if response = C

then:

```
CONCENTRATION (GRAMS/LITER) = ?
NUMBER OF SIZE FRACTIONS (1 to 9) = ?
% OF MASS FOR FRACTION 1 ( 0 TO 100) = ?
SETTLING VELOCITY FOR FRACTION 1 (CM/SEC) = ?
```

repeat until data is obtained for the number of fractions specified.

Each of the questions is answered by inputting the appropriate information followed by a carriage return. FRNT generates a file called PLUME.VRS (plume-variables). This file looks like:

```
 1.  U        =     0.05
 2.  H        =      100
 3.  FF       =     0.01
 4.  DELX     =     1000
 5.  DELZ     =      100
 6.  NXS      =       41
 7.  NYS      =        5
 8.  NZS      =        5
 9.  2*BO     =       50
10.  WBOT     =        3
11.  WTOP     =        4
12.  OPT      =        M
13.  FLUX     =   100000
14.  NFACS    =        1
15.  PCTMF    =      100
16.  V        =        0
```

Entries in the last column are responses which the modeler gave to FRNT. The order of these entries is the same as that of the questions which FRNT asks. The file PLUME.VRS may be edited, copied to the printer, etc. In editing the file, the last column must be properly right justified.

Running the Model

16. To run the plume model, the modeler simply types in MODEL. The computer will first output the relevant input variables to the printer, and then respond with a series of questions which are as follows:

PRINT TABLE OF MASS FLUXS (Y OR N)?
PRINT SUM OF CO TABLE (Y OR N)?
PRINT TABLE OF CD VALUES (Y OR N)?
PRINT CONCENTRATION TABLE (Y OR N)?

To generate a table of concentrations

ENTER THE DEPTH AT WHICH CONCENTRATION IS WANTED

ro stop generating concentration tables

ENTER A NEGATIVE NUMBER
(DEPTHS ARE ROUNDED TO THE NEAREST NODAL EVALUATION)

PRINT TABEL OF DEPOSITION RATES (Y OR N)?

The tables corresponding to the second and third questions are the solutions (if one fraction is specified) of Equations 8 and 9 respectively. If more than 1 fraction is specified, the second table is the solution of Equation 8 summed over all of the fractions. The table corresponding to the first question gives the total mass flux and the mass flux for the various fractions across planes normal to the x-axis at various downstream locations. Concentrations at various depths are generated in response to the fourth query, and as many of these tables as are desired may be generated. Entry of a negative number halts the generation of these tables. The last table which may be generated is a table of deposition rates. This table gives deposition rates as mass per unit area per unit time.

Special Considerations

17. Referring back to Equation 13 in which vertical diffusion is ignored, observe that in the time it takes a particle to advance a distance dx, the distance between two stations in the longitudinal direction, it will fall

a distance

$$- dy = w/U \, dx \tag{17}$$

Equation 14 uses two succesive grid elevations under the assumption that the characteristic line

$$y = w x / U \tag{18}$$

falls in the interval $y(j)$ to $y(j+1)$. This will happen only if the distance dh between two vertical stations is greater than or equal to dy . The inequality consistent with the assumption is

$$dh > dy = w/U \, dx \tag{19}$$

If this inequality is violated the numerical solution of Equation 8 becomes unstable (wild unpredictable numerical output is generated). Since the depth h , the settling velocity w , and the mean current velocity are fixed, and since

$$dh = h / (Nys-1) , \tag{20}$$

where Nys is the number of stations in the vertical direction, Equation 19 imposes a restriction on the values of dx and Nys which may be used. The most restrictive case in a given situation will be for the material with the greatest settling velocity. The computer checks the inequality and when it is not satisfied prints the message

```
Dy/Dx MUST BE > OR = w/U FOR STABILITY
w/U = .........  DELY/DELX = .........
DECREASE DELX OR NYS ... COMPUTATION ABORTED
```

In a number of instances it has been found desirable to run the model first with a relatively small dx until the sand fraction settles out. After removing the sand fraction from the distribution the model is rerun with a correspondingly larger dx value.

18. Suppose that there is one fraction with zero settling velocity. Material must be conserved within the flow and the mass flux across any section normal to the x-axis will be

$$\text{mass-flux} = (2boh)\ avg(C_o)U \tag{21}$$

This equation is satisfied by the numerical model if the equation

$$\mathrm{avg}(C_o) = \sum_{n=2}^{Nvs-1} C_n \tag{22}$$

is used to compute the average of C_o. In many instances it is the mass flux and not the initial concentration which is known. The numerical model uses Equations 21 and 22 to compute the upstream concentration when the mass flux is given. Using the divergence theorem it is not difficult to show that the mass flux per unit area across a horizontal surface is

$$q = wC + E_y \frac{\partial c}{\partial y} \tag{23}$$

At the bottom the boundary condition is

$$\partial C/\partial y = 0 \tag{24}$$

Hence, the rate of deposition is just

$$q = wC \tag{25}$$

PART IV: SUMMARY AND CONCLUSIONS

Summary

19. The plume model developed in this study incorporates the major physical mechanisms which influence the behavior of a suspended material. However, the model is still simple enough to be run on relatively small computers. The model does, however, have limited applicability because of the many assumptions made, e.g. the current is considered independent of time and is constant in magnitude and direction.

20. The model has been run on both 8 bit micro and 32 bit super microcomputers. The modeling system consists of two programs, FRNT which sets the problem up, and MODEL which performs the actual computations. Both of the programs are interactive and use a question and response mechanism to communicate with the user. Variables such as the current velocity are fixed by physical circumstances while other variables such as the number of stations in the x, y, and z directions are free for the modeler to choose. The only restriction on the choice of these variables is the stability criterion indicated in PART III of this report. The computer code (MODEL) checks Equation 19 and aborts the run if the criterion is not met.

21. A number of runs have been made in testing the model. The coarsest material (and the size of the computer memory) limit the longitudinal distance which can be modeled because of the stability criterion. When both sand and clay material are present, the plume length which can be modeled gives an unsatisfactory picture of the fate of the finer material. What has proved to be useful is to remove the coarser fraction from the distribution and repeat the computation with an appropriately larger distance between longitudinal stations. This practice is acceptable because the various fractions are assumed to behave independently in the model.

Conclusions

22. Much more sophisticated models of sediment diffusion processes are available. However, these models require a substantial body of field information which must be obtained for each specific problem, and also require a substantial investment in manpower and large computer costs for calibration

and testing. In some instances the data simply aren't available and/or the cost of obtaining them and applying a sophisticated model is unwarranted. The model described in this report gives a realistic general picture of plume behavior and should be particularly useful when the use of more sophisticated models is impractical.



FIGURE 1. DEFINITION SKETCH

Appendix A: Listing of Model

```
      INTEGER FORMFD,LF
      INTEGER*1 OPT,YORN
      DOUBLE PRECISION D(21,21),USTR,FX,FNYS2,UDX,E,EM,EP,B(21)
      DIMENSION CD(61,11),C(21),FLUX(6,61),V(5),CO(21,61),COT(21,61)
      DIMENSION VCOB(61,6),NUM(10)
      DIMENSION DPTH(21),XDIS(61),ZDIS(11)
      DATA FORMFD/12/,CD/441*0./,COT/441*0.0/,LF/10/

   11 FORMAT(19X,F10.0)
    1 FORMAT(20X,F10.0)
    2 FORMAT(20X,I10)
    3 FORMAT(29X,A1)
    4 FORMAT(' FRACT. NUM. ',I2,' % OF MASS = ',F4.0,
     1        'FALL VELOCITY = ',F10.5)

      DO 10 N = 1,10
   10 NUM(N)=N
C
      WRITE(2,6030) FORMFD
C
      CALL OPEN (6,11HPLUME   VRS,1)
      READ(6,11)U
      READ(6,1)H
      READ(6,1)FR
      WRITE(2,5)U,FR,H
    5 FORMAT(' U = ',F10.5,' FR = ',F10.5,' H = ',F10.5)
      READ(6,1)DELX
      READ(6,1)DELZ
      WRITE(2,6)DELX,DELZ
    6 FORMAT(' DELX = ',F10.1,' DELZ = ',F10.1)

      READ(6,2)NXS
      READ(6,2)NYS
      READ(6,2)NZS
      WRITE(2,7)NXS,NYS,NZS
    7 FORMAT(' NXS = ',I4,' NYS = ',I4,' NZS = ',I4)

      READ(6,1)TWOBO
                 BO = TWOBO/2.
                 BO = BO
      WRITE(2,8)TWOBO
      READ(6,2)IBOTW
      WRITE(2,9)IBOTW
      READ(6,2)ITOPW
      WRITE(2,13)ITOPW
    8 FORMAT(' WIDTH OF WINDOW = ',F10.1)
    9 FORMAT('       Y NODE BOTTOM OF WINDOW = ',I5)
   13 FORMAT('       Y NODE AT TOP OF WINDOW = ',I5)

      READ(6,3)OPT
      WRITE(2,15)OPT,OPT
   15 FORMAT(' OPT = ',A1,2X,I3)
      IF(OPT .NE. 77) GO TO 20
      READ(6,1)FLUXO
      FNYS=NYS-1
      DELH=H/FNYS
      IF(IBOTW.EQ.1)IBOTW=2
      IF(ITOPW.EQ.NYS)ITOPW=NYS-1
      HW= FLOAT(ITOPW-IBOTW)/FNYS
      IF(HW.LT.0.5)HW=1
```

```
      HW=HW*DELH
      AW=TWOBO*HW
      CONCO=FLUXO/(U*AW)
      WRITE(2,19)CONCO
   19 FORMAT(' CONC. AT U_S BOUNDARY = ',1PE12.3)
      FNYS=FLOAT(NYS-2)
      RATIO = FLOAT(ITOPW-IBOTW+1)/FNYS
      CONC = FLUXO/(U*TWOBO*H*RATIO)
      GO TO 30

   20 READ(6,1)CONC

C
   30 CONTINUE
      XDIS(1)=0.
      DO 32 I = 2,NXS
   32 XDIS(I)=XDIS(I-1)+DELX
      ZDIS(1)=0.
      DO 34 I = 2,NZS
   34 ZDIS(I)=ZDIS(I-1)+DELZ
      FNYS1= NYS-1
      FNYS2= NYS-2
      DELH=H/FNYS1
      DELY=DELH
      DPTH(NYS)=0.
      J=NYS
      DO 36 I = 2,NYS
      J=J-1
      DPTH(J)=DPTH(J+1) + DELH
   36 CONTINUE

      WRITE(2,21)CONC
   21 FORMAT(' GROSS INITIAL CONCENTRATION = ',F10.2)
C
C
      READ(6,2)NFRACS
      WRITE(2,23)NFRACS
   23 FORMAT(' NUMBER OF MASS FRACTIONS = ',I2)
C
C
      DO 99 I=1,NYS
      DO 99 J=1,NXS
   99 COT(I,J)=0.

      DO 1000 L=1,NFRACS
         READ(6,1)PCNTMS
         READ(6,1)V(L)
         V(L)=V(L)/100.
C CONVERSION FRM CM/SEC TO M/SEC
                IF(V(L).GT.0)V(L)=-V(L)
                WRITE(2,4)L,PCNTMS,V(L)
                CONCFR =CONC*(PCNTMS/100.)
        NYS1=NYS-1
        DO 60 I =1,NYS
        IF(I.GE.IBOTW)GO TO 52
        CO(I,1)=0.
        GO TO 60
   52   IF(I.LE.ITOPW)GO TO 54
        CO(I,1)=0.
        GO TO 60
```

```
   54    CO(I,1)=CONCFR
   60    CONTINUE
C
C234567
       VRATIO=ABS(V(L))/U
       FNYS1 = NYS-1
       DELY  = H/FNYS1
       GRATIO=DELY/DELX
       IF(GRATIO.GE.VRATIO)GO TO 70
       WRITE(2,61)VRATIO,GRATIO
   61  FORMAT(' DELY/DELX MUST BE > OR = V/U FOR STABILITY '/
      1  ' V/U = ',F10.5,' DELY/DELX = ',F10.5,/
      2  ' DECREASE DELX OR  NYS... COMPUTATION ABORTED ')
       STOP
   70  CONTINUE
          USTR=SQRT(FR/8.)*U
          FX=0.4*USTR
          FNYS1=NYS-1
          FNYS2=NYS-2
          FNYS2=1./FNYS2
          DELY=H/FNYS1
          HDELY=0.5*DELY
          VDY=0.5*V(L)/DELY
          UDX=U/DELX
          SDYI=1./DELY**2
          DYI=1./DELY
          RATIO=(V(L)*DELX)/(U*DELY)
C
          DO 80 I=1,NYS
          DO 80 J=1,NYS
             D(I,J)=0.
    80    CONTINUE
C
          NYS1=NYS-1
C
          D(1,1)=-DYI
          D(1,2)=DYI
C
          DO 100 N=2,NYS1
               Y=(N-1)*DELY
               YM=Y-HDELY
               YP=Y+HDELY
               EM=FX*U*YM*(1.-YM/H)
               EP=FX*U*YP*(1.-YP/H)
               E=0.5*(EM+EP)
               D(N,N-1)=-EM*SDYI
               D(N,N)=UDX+2.*E*SDYI
               D(N,N+1)=-EP*SDYI
   100    CONTINUE
C
          D(NYS,NYS)=V(L)-EP*DYI
          D(NYS,NYS1)=+EP*DYI
C
          UDXI=1./UDX
          D(1,1)=UDXI*D(1,1)
          D(1,2)=UDXI*D(1,2)
C
          DO 200 I=2,NYS1
               D(I,I-1)=UDXI*D(I,I-1)
               D(I,I)=UDXI*D(I,I)
```

```
                 D(I,I+1)=UDXI*D(I,I+1)
   200      CONTINUE
C
            DO 300 I=2,NYS
                 D(I,I-1)=D(I,I-1)/D(I-1,I-1)
                 D(I,I)=D(I,I)-D(I,I-1)*D(I-1,I)
   300      CONTINUE
C
C
            N=1
            SUM=0.0
C
            DO 350 J=2,NYS1
   350      SUM=SUM+CO(J,N)
            FLUX(L,N)=2.*U*BO*H*SUM*FNYS2
C
            DO 800 N=2,NXS
                 B(1)=0.
                 DO 400 I=2,NYS1
                      B(I)=CO(I,N-1)-RATIO*(CO(I+1,N-1)-CO(I,N-1))
   400           CONTINUE
C
                 B(NYS)=0.
                 DO 500 I=2,NYS
                      B(I)=B(I)-D(I,I-1)*B(I-1)
   500           CONTINUE
C
                 B(NYS)=B(NYS)/D(NYS,NYS)
                 I=NYS
                 DO 600 K=2,NYS
                      I=I-1
                      B(I)=(B(I)-D(I,I+1)*B(I+1))/D(I,I)
   600           CONTINUE
C
                 DO 700 I=1,NYS
   700           CO(I,N)=B(I)
C
        VCOB(N,L)=-V(L)*CO(1,N)
C VCOB MAY BE THOUGHT OF AS HAVING THE DIMENSIONS
C GRAMS/SQ-METER PER SECOND IF ONE WILL REGARD
C CD AS DIMENSIONLESS

                 SUM=0.0
                 DO 750 J=2,NYS1
                    SUM=SUM+CO(J,N)
   750           CONTINUE
C
                 FLUX(L,N)=2.*U*BO*H*SUM*FNYS2
C
   800      CONTINUE
C
            DO 820 I=1,NYS
            DO 820 J=1,NXS
            COT(I,J)=COT(I,J)+CO(I,J)
   820      CONTINUE
C
  1000 CONTINUE
C
        WRITE(3,881)
  881   FORMAT(' PRINT TABLE OF MASS FLUX <Y OR N > ? ')
```

```
      READ(3,882)YORN
  882 FORMAT(A1)
      IF(YORN .EQ. 78)GO TO 895
      WRITE(2,6030)FORMFD
      WRITE(2,883)
      WRITE(2,507)
      WRITE(2,508)
      WRITE(2,501)(NUM(I),I=1,NFRACS)
  883 FORMAT(5X,' TOTAL MASS FLUX AND FLUX FOR FRACTIONS ')
  507 FORMAT(5X,'                    GRAMS/SEC '/)
  508 FORMAT('    X       TOTAL    FRACTION NUMBER ')
  501 FORMAT(' METERS   FLUX       ',I5,5I10)
  884 FORMAT('  X(METERS) ')
      DO 890 N=1,NXS
         SUM=0.
         DO 870 L=1,NFRACS
             SUM=SUM+FLUX(L,N)
  870    CONTINUE
C
         WRITE(2,887)XDIS(N),SUM,(FLUX(L,N),L=1,NFRACS)
  887 FORMAT(1X,F8.0,1PE10.2,1PE10.2,1PE10.2,1PE10.2,1PE10.2,
     1        1PE10.2,1PE10.2,1PE10.2,1PE10.2,1PE10.2,1PE10.2)
  890 CONTINUE
  895 CONTINUE
C
C234567
      WRITE(3,891)
  891 FORMAT(' PRINT SUM OF CO TABLE <Y OR N > ? ')
      READ(3,882)YORN
      IF(YORN .EQ. 78)GO TO 955
      JDEL=(NYS-1)/4
      JYS=1+JDEL*4
      IF (JYS .EQ. NYS) GO TO 899
      WRITE(3,951)
  951 FORMAT(' NO. OF Ys MUST EQUAL 5,9,13... TO PRINT THIS TABLE ')
      GO TO 955

  899 WRITE(2,6030)FORMFD
C
      WRITE(2,931)
  931 FORMAT(20X,'TABLE OF CO VALUES (MILLIGRAMS/LITER) '/)
      WRITE(2,932)(DPTH(I),I=1,NYS,JDEL)
  932 FORMAT(' DEPTH = ',F7.0,4F10.0)

      WRITE(2,884)
      DO 950 I =1,NXS
      WRITE(2,953)XDIS(I),(COT(J,I),J=1,NYS,JDEL)
  953 FORMAT(1X,F8.0,1PE10.2,1PE10.2,1PE10.2,1PE10.2,1PE10.2)
  950 CONTINUE
  955 CONTINUE
      WRITE (3,957)
  957 FORMAT (' COMPUTING CD, PLEASE WAIT '//)

      CD(1,1)=1.0
      DO 1050 I = 2,NZS
 1050 CD(I,I)=0.0

      EZ=0.2*H*USTR
      DO 1200 N=2,NXS
         X=(N-1)*DELX
```

```
              AL=SQRT(U/(4*X*EZ))
 C
              DO 1100 J=1,NZS
                    Z=(J-1)*DELZ
                    ARG1=AL*(BO+Z)
                    ARG2=AL*(BO-Z)
                    ERF1=ERF(ARG1)
                    IF (ARG2 .LT. 0.) GO TO 441
                    ERF2=ERF(ARG2)
                    CD(N,J)=0.5*(ERF1+ERF2)
                    GO TO 1100
  441              ERF2=ERF(-ARG2)
                    CD(N,J)=0.5*(ERF1-ERF2)
  1100        CONTINUE
 C
  1200 CONTINUE
       WRITE(3,1201)
  1201 FORMAT(' PRINT TABLE OF CD VALUES <Y OR N> ? ')
       READ(3,882)YORN
       IF(YORN.EQ.78)GO TO 1300
       WRITE(2,6030)FORMFD
       WRITE(2,1202)
  1202 FORMAT(20X,' TABLE OF CD VALUES ')
       WRITE(2,1203)(ZDIS(I),I=1,NZS)
  1203 FORMAT('  Z = ',11F10.0)
       WRITE(2,884)
       DO 1250 I = 1,NXS
       WRITE(2,887)XDIS(I),(CD(I,J),J=1,NZS)
  1250 CONTINUE
  1300 CONTINUE
  6030 FORMAT (A4)
 C
       WRITE(3,1301)
  1301 FORMAT(' TO GENERATE A TABLE OF CONCENTRATTION: '/
      1       ' ENTER THE DEPTH AT WHICH CONCENTRATION IS WANTED '/
      2       ' TO STOP GENERATING CONCENTRATION TABLES: '/
      3       ' ENTER A NEGATIVE NUMBER '//
      4       ' <DEPTHS ARE ROUNDED TO THE NEAREST NODAL ELEVATION> '/)
  1320 CONTINUE
       WRITE(3,1302)
  1302 FORMAT(' DEPTH IN METERS = ? ')
       READ(3,1303)DEPTH
  1303 FORMAT(F10.0)
       IF(DEPTH .LT. 0.)GO TO 1400
       WRITE(2,6030)FORMFD
       IY = DEPTH/DELH
       REM = DEPTH -FLOAT(IY)*DELH
       IF(REM.GT.0.5*DELH)IY=IY+1
       DEPTH = FLOAT(IY)*DELH
       WRITE(2,1304)
  1304 FORMAT(20X,' TABLE OF CONCENTRATIONS (MILLIGRAMS/LITER) ')
       WRITE(2,1305)DEPTH
       WRITE(2,1203)(ZDIS(I),I=1,NZS)
       WRITE(2,884)
       IY = NYS - IY
       DO 1380 N = 1,NXS
       DO 1340 I = 1,NZS
  1340 C(I) = COT(IY,N)*CD(N,I)
       WRITE(2,887)XDIS(N),(C(I),I=1,NZS)
  1380 CONTINUE
```

```
 1305 FORMAT(25X,' AT A DEPTH OF ',F5.0,' METERS '/)
      GO TO 1320
 1400 CONTINUE
      WRITE(3,1401)
 1401 FORMAT(' PRINT TABLE OF DEPOSITION RATES <Y OR N> ? ')
      READ(3,882)YORN
      IF(YORN.EQ.78)GO TO 1500
      WRITE(2,6030)FORMFD
      WRITE(2,1402)
 1402 FORMAT(20X,' TABLE OF DEPOSITION RATES ')
      WRITE(2,1403)
 1403 FORMAT(20X,'   (GRAMS/SEC PER M-SQ )'/)
      WRITE(2,1203)(ZDIS(J),J=1,NZS)
      WRITE(2,884)
      DO 1460 I = 1,NXS
      SUM =0.
      DO 1420 K = 1,NFRACS
      SUM = SUM+VCOB(I,K)
 1420 CONTINUE
      DO 1440 J = 1,NZS
 1440 C(J)=SUM*CD(I,J)
C IN THIS CONTEXT C(J) IS A CONVENIENT DUMMY
      WRITE(2,887)XDIS(I),(C(J),J=1,NZS)
 1460 CONTINUE
 1500 CONTINUE
      STOP
      END
C
      FUNCTION ERF(X)
C
      DOUBLE PRECISION A1,A2,A3,A4,A5,T,P
      DATA A1,A2,A3 /.254829592,-.284496736,1.421413741/
      DATA A4,A5    /-1.453152027,1.061405429/
C
      EXPF=0.0
      IF (X .GT. 4.2) GO TO 10
      T=1./(1.+.3275911*X)
      P=((((A5*T+A4)*T+A3)*T+A2)*T+A1)*T
      EXPF=EXP(-X*X)
   10 ERF=1.-P*EXPF
C
          RETURN
          END
```

Appendix B: Listing of FRNT

```
      PROGRAM FRNT
      INTEGER*1 FLD(20)
      INTEGER*1 CR,LF,ZERO
      INTEGER*1 VN
      CR=13
      LF=10
      ZERO = 48

      CALL OPEN(7,11HPLUME   VRS,1)

      WRITE(3,3)
      CALL GETFLD(FLD)
    3 FORMAT(' STREAM VELOCITY (M/SEC) = ? ')
      WRITE(7,4)(FLD(I),I=1,20)

      WRITE(3,7)
    7 FORMAT(' DEPTH IN METERS = ? ')
      CALL GETFLD(FLD)
      WRITE(7,8)LF,(FLD(I),I=1,20)

      WRITE(3,9)
    9 FORMAT( ' FRICTION FACTOR (RANGE = 0.0005 TO 0.025) = ? ')
      CALL GETFLD(FLD)
      WRITE(7,10)LF,(FLD(I),I=1,20)

      WRITE(3,11)
   11 FORMAT(' LONGITUDIONAL DIST. BETWEEN STATIONS IN METERS = ? ')
      CALL GETFLD(FLD)
      WRITE(7,12)LF,(FLD(I),I=1,20)

      WRITE(3,13)
   13 FORMAT(' LATERAL DIST. BETWEEN STATIONS IN METERS = ? ')
      CALL GETFLD(FLD)
      WRITE(7,14)LF,(FLD(I),I=1,20)

      WRITE(3,15)
   15 FORMAT(' NUMBER OF STATIONS IN LONGITUDIONAL DIRECTION = ? ')
      CALL GETFLD(FLD)
      WRITE(7,16)LF,(FLD(I),I=1,20)

      WRITE(3,17)
   17 FORMAT(' NUMBER OF STATIONS IN THE VERTICAL DIRECTION = ? ')
      CALL GETFLD(FLD)
      WRITE(7,18)LF,(FLD(I),I=1,20)

      WRITE(3,19)
   19 FORMAT(' NUMBER OF STATIONS IN THE LATERAL DIRECTION = ? ')
      CALL GETFLD(FLD)
      WRITE(7,24)LF,(FLD(I),I=1,20)

      WRITE(3,27)
   27 FORMAT(' WIDTH OF INFLOW WINDOW (METERS) = ? ')
      CALL GETFLD(FLD)
      WRITE(7,38)LF,(FLD(I),I=1,20)

      WRITE(3,29)
   29 FORMAT(' STATION NUMBER FOR BOTTOM OF WINDOW ( 1 IS BOTTOM) = ? ')
      CALL GETFLD(FLD)
      WRITE(7,34)LF,(FLD(I),I=1,20)
```

```
      WRITE(3,31)
   31 FORMAT(' STATION NUMBER FOR TOP OF WINDOW ( =< NUMB. OF Ys) = ? ')
      CALL GETFLD(FLD)
      WRITE(7,36)LF,(FLD(I),I=1,20)

      WRITE(3,33)
   33 FORMAT(' MASS FLUX OR CONC. BOUN_COND (M OR C) ? ')
      CALL GETFLD(FLD)
      WRITE(7,42)LF,(FLD(I),I=1,20)
      VN = FLD(20)

      IF(VN .EQ. 77)GO TO 80
      WRITE(3,35)
   35 FORMAT( ' CONCENTRATION (GRAMS/LITER) = ? ')
      CALL GETFLD(FLD)
      WRITE(7,44)LF,(FLD(I),I=1,20)

      GO TO 90
   80 CONTINUE

      WRITE(3,21)
   21 FORMAT(' MASS FLUX (IN UNITS OF GRAMS/SEC) = ? ')
      CALL GETFLD(FLD)
      WRITE(7,26)LF,(FLD(I),I=1,20)

   90 CONTINUE

      WRITE(3,23)
   23 FORMAT(' NUMBER OF SEDIMENT SIZE FRACTIONS (1 TO 9) = ? ')
      CALL GETFLD(FLD)
      WRITE(7,28)LF,(FLD(I),I=1,20)

      NUM = FLD(20)-ZERO

      DO 100 NN = 1,NUM
      WRITE(3,25),NN
   25 FORMAT(' % OF MASS  FOR FRACTION',I2,' (0 TO 100) = ? ')
      CALL GETFLD(FLD)
      WRITE(7,32)LF,(FLD(I),I=1,20)

      WRITE(3,5)NN
    5 FORMAT(' SETTLING VELOCITY FOR FRACTION',I2,' (CM/SEC) = ? ')
      CALL GETFLD(FLD)
      WRITE(7,6)LF,(FLD(I),I=1,20)


  100 CONTINUE

C                  12345678
    4 FORMAT(' U      = ',20A1)
    6 FORMAT(A1,'     V  = ',20A1)
    8 FORMAT(A1,' H      = ',20A1)
   10 FORMAT(A1,' FF     = ',20A1)
   12 FORMAT(A1,' DELX   = ',20A1)
   14 FORMAT(A1,' DELZ   = ',20A1)
   16 FORMAT(A1,' NXS    = ',20A1)
   18 FORMAT(A1,' NYS    = ',20A1)
   24 FORMAT(A1,' NZS    = ',20A1)
   26 FORMAT(A1,' FLUX   = ',20A1)
   28 FORMAT(A1,' NFRCS = ',20A1)
```

```
 32   FORMAT(A1,' PCTMF = ',20A1)
 34   FORMAT(A1,' WBOT  = ',20A1)
 36   FORMAT(A1,' WTOP  = ',20A1)
 38   FORMAT(A1,' 2*BO  = ',20A1)
 42   FORMAT(A1,' OPT   = ',20A1)
 44   FORMAT(A1,' CONC  = ',20A1)
 46   FORMAT(A1)

      WRITE(7,46)LF
      STOP
      END
      SUBROUTINE GETFLD(FLD)
      INTEGER*1 CHARS(80),FLD(20),SP
      READ(3,1)(CHARS(I),I=1,80)
  1   FORMAT(A1,80A1)
        NCHARS = 80
      CALL COUNT(CHARS,NCHARS)
      CALL FIELD(CHARS,FLD,NCHARS)
      RETURN
      END
      SUBROUTINE COUNT(CHARS,NCHARS)
      INTEGER*1  CHARS(80),SP
      SP = 32
      NCHARS = 80
      DO 20 I = 1,80
      NCHARS = NCHARS - 1
      IF(CHARS(NCHARS).NE.SP)GO TO 40
 20   CONTINUE
 40   CONTINUE
      RETURN
      END
      SUBROUTINE FIELD(CHARS,FLD,NCHARS)
      INTEGER*1  CHARS(80),FLD(20)
      INTEGER*1  SP
      SP = 32
      NSPS = 20 - NCHARS
      DO 20 I = 1,NSPS
      FLD(I) = SP
 20   CONTINUE
      DO 40 I = 1,NCHARS
      N = I + NSPS
      FLD(N) = CHARS(I)
 40   CONTINUE
      RETURN
      END
```

The Department of the Interior Mission

As the Nation's principal conservation agency, the Department of the Interior has responsibility for most of our nationally owned public lands and natural resources. This includes fostering sound use of our land and water resources; protecting our fish, wildlife, and biological diversity; preserving the environmental and cultural values of our national parks and historical places; and providing for the enjoyment of life through outdoor recreation. The Department assesses our energy and mineral resources and works to ensure that their development is in the best interests of all our people by encouraging stewardship and citizen participation in their care. The Department also has a major responsibility for American Indian reservation communities and for people who live in island territories under U.S. administration.

The Minerals Management Service Mission

As a bureau of the Department of the Interior, the Minerals Management Service's (MMS) primary responsibilities are to manage the mineral resources located on the Nation's Outer Continental Shelf (OCS), collect revenue from the Federal OCS and onshore Federal and Indian lands, and distribute those revenues.

Moreover, in working to meet its responsibilities, the **Offshore Minerals Management Program** administers the OCS competitive leasing program and oversees the safe and environmentally sound exploration and production of our Nation's offshore natural gas, oil and other mineral resources. The MMS **Minerals Revenue Management** meets its responsibilities by ensuring the efficient, timely and accurate collection and disbursement of revenue from mineral leasing and production due to Indian tribes and allottees, States and the U.S. Treasury.

The MMS strives to fulfill its responsibilities through the general guiding principles of: (1) being responsive to the public's concerns and interests by maintaining a dialogue with all potentially affected parties and (2) carrying out its programs with an emphasis on working to enhance the quality of life for all Americans by lending MMS assistance and expertise to economic development and environmental protection.